
Abstract. This paper reports on some prelimary research into how
software tools like InterLoc can be used as an interface to the World
Wide Argument Web (WWAW) and how the WWAW in return can
provide a useful resource to agents acting within InterLoc. Two
persuasive dialogue games, the human-oriented Critical Reasoning
Game (CRG) from InterLoc and the philosophy-based agent-oriented
game for permissive persuasion named PPD0 are compared using
the Dialogue Game Description Language (DGDL) as an interlingua.
The expressiveness of each game is investigated by mapping output
dialogues onto argumentation structures represented in the Argument
Interchange Format (AIF).

1 INTRODUCTION

This paper reports on preliminary research into developing persua-
sive online software systems that integrate naturalistic human di-
alogues, thus spurring increased user engagement, with formally
structured argumentation, supporting automated processing by intel-
ligent agents and interconnection of resources online.

InterLoc [4] is software that can be used to support dialogues be-
tween groups of users and enable them to interactively explore a topic
domain. This has been used thus far in a primarily educational con-
text to facilitate debate between students. This approach can be ex-
tended through the use of intelligent tutor agents to enable new do-
main knowledge to be introduced into the student dialogues enabling
the students to explore various paths through the topic and increasing
their knowledge. Tutor agents could also be used to scaffold and di-
rect the dialogues so that important topics were covered in sufficient
depth or to ensure that the dialogue was steered towards the conclu-
sions that the tutor wishes for the students to discover. Additionally,
intelligent conversational agents could play the role of a devils advo-
cate in an adversarial dialogue in which the students defend a given
position based upon their knowledge of the domain or in which the
agent tries to persuade the students to accept a position that differs
from their starting position.

Recently there has also been increasing interest in online argumen-
tation, for example MAgtALO [5] provides an interface for human-
agent dialogue whereas ArgDF [3] provides an interface for con-
structing arguments using argumentation schemes [6]. Some aspects
of online argumentation systems, for example the user facing inter-
faces like that in MAgtALO, suggest a good way to usefully deploy
well-reasoned persuasive argumentation. By adopting natural inter-
faces, that support naturalistic human interaction, users can explore

a problem domain, and can be supported and guided towards well-
reasoned conclusions, a form of gentle persuasion technology rather
than “hard sell” persusasion.

There has also been broad interest in the underlying formal rep-
resentations that support the widespread sharing and interchange
of argumentative resources in online systems. This is useful, not
only to support the development and deployment of persuasive
argumentation-based interfaces, but also to support more advanced
online argumentation processing. Work towards this end has culmi-
nated in the nascent Argument Interchange Format (AIF) [1] used to
record and share argument resources and a foundational element of
the proposed World Wide Argument Web (WWAW) [2].

Initially our research has investigated the argumentative structures
that can be extracted from InterLoc dialogues. Our aim in this task
is to investigate the suitability of InterLoc as an interface to the
WWAW, both as a means to elicit new arguments into the system,
but also as a way for humans to explore existing WWAW argument
resources using a naturalistic interface. Further to this we have in-
vestigated dialogue games from argumentation theory that are useful
for implementing agent argumentation, in particular we have investi-
gated dialogue game protocols that can be used to regulate persuasion
dialogues according to the criteria of Walton and Krabbe [7].

The aim is to be able to incorporate argumentative intelligent
agents into WWAW interfaces, such as InterLoc, without compro-
mising the human friendly aspects of the current InterLoc interac-
tion protocols. To achieve this we aim to balance the more expres-
sive dialogue protocols which support naturalistic human dialogical
interaction against the more formally structured protocols that are
used in intelligent conversational agents. By doing this we propose
the construction of protocols that are sufficiently expressive to allow
naturalistic human interaction without introducing significant cogni-
tive overhead but which are also sufficiently structured and formally
underpinned to support support agent interaction.

2 TECHNOLOGIES

This research integrates a number of extant technologies from the
domains of educational software and argumentation theory. In our
preliminary work we have drawn together a range of theoretical tools
which we are exploring with the aim of assembling them into a co-
hesive software architecture to meet the goals discussed in section 1.
Our exploration of technologies has thus far been confined to evalu-
ating two dialogue games, the Critical Reasoning Game (CRG) [4]
and the Permissive Persuasion Dialogue game (PPD0) [7]. To support
the comparison and analysis of these games other technologies have
been adopted, the Dialogue Game Description Language (DGDL)
for describing disparate protocols using a common language, and the



Argument Interchange Format (AIF) for representing the outcome
dialogues produced from the dialogue games.

CRG is an interaction protocol used in InterLoc. This is a human-
oriented dialogue game based interaction protocol which specifies
a permissive, free-ranging dialogue between numerous locutors. An
advantage of this protocol is that it is flexible, expressive and permis-
sive and scaffolds dialogues by suggesting ways that the dialogue can
develop rather than ways that it must develop. However this syntac-
tical permissiveness and the lack of a semantic model provides little
structure with which to support an intelligent agent interacting with
the human locutors in an InterLoc dialogue.

Dialogue games used in agent communication typically incor-
porate commitment models. One such dialogue game developed to
model permissive persuasion dialogues is PPD0 which incorporates
commitment stores which are used to track the commitments of play-
ers with respect to the locutions uttered during a dialogue. In PPD0

the legal sequences of locutions are defined both in terms of the set
of locutions that may follow an earlier locution and also the com-
mitment state of the players. Because of these kinds of rule PPD0 is
relatively heavyweight and restrictive in comparison to CRG but this
means that at any given point in a PPD0 dialogue the set of things
that can be said, and therefore the set of alternative utterances that an
agent must select from, is much smaller.

Both CRG and PPD0 are, in their original formulations in [4] and
[7] respectively, specified using different mechanisms. CRG is ex-
pressed using XML and PPD0 is expressed in a natural language
description. To aid in the comparison of the games, and to reduce
the complexity of transcribing rules from one game to another, the
Dialogue Game Description Language (DGDL) [8] was used as an
interlingua. DGDL is a domain specific language for describing di-
alogue games whose syntax is underpinned by an EBNF grammar.
This supports the rapid development of syntactically correct dialogue
game descriptions that can be deployed in agent software.

3 THE DIALOGUE GAME DESCRIPTION
LANGUAGE (DGDL)

Communication is an important topic within intelligent agent re-
search and is a fundamental factor in the development of robust
and efficient multiagent systems. Similarly, argumentation has been
recognised as a key component of an agents ability to make decisions
using complex, dynamic, uncertain, and incomplete knowledge. Di-
alectical games are a type of multi-player argumentative dialogue
game and provide a mechanism for communication which incorpo-
rates argumentative behaviours. However there have been very few
tools for working with these games and little agreement over how
they should best be described, shared, and reused. The Dialogue
Game Description Language (DGDL) [8] is a domain specific lan-
guage for describing dialectical games and provides a grammar for
determining whether a game description is syntactically correct and
thus provides a foundation for new tools to support the future devel-
opment and wider exploitation of dialectical games.

The DGDL grammar supports the syntactically correct descrip-
tion of a wide array of dialectical games, whether extant games or
wholly new formulations of rules. Games are described in terms of
their composition, including specification of participants, turn struc-
ture, and commitment stores, their rules, regulations that manipulate
the game components indirectly, and their interactions, the moves
that players can make that directly manipulate game components.

To support the comparison of PPD0 and CRG, the original natu-
ral language description of rules was formalised into a DGDL game

description as follows:

PPD0{
{turns, magnitude:multiple, ordering:strict };
{ roles, {Speaker, Listener} };
{players, id:black, roles:{ Speaker } };
{players, id:white, roles:{ Listener } };
{store, id:Assertions, owner:black, structure:set, visibility:public };
{store, id:Assertions, owner:white, structure:set, visibility:public };
{store, id:Concessions, owner:black, structure:set, visibility:public };
{store, id:Concessions, owner:white, structure:set, visibility:public };
{store, id:Dark, owner:black, structure:set, visibility:private };
{store, id:Dark, owner:white, structure:set, visibility:private };

{Commencement, scope:initial,
{ move(mandate, next, Assertion, Speaker) } };

{SpeakerWins, scope:turnwise,
{ if { inspect(in,{p},Assertions,Listener,initial)
& inspect(!in,{p},Assertions,Listener,current) }
then { status(terminate,PPD0), assign(speaker, winner) } } };

{ListenerWins, scope:turnwise,
{ if { inspect(in,{p},Assertions,Speaker,initial)
& inspect(!in,{p},Assertions,Speaker,current) }
then { status(terminate,PPD0), assign(listener, winner) } } };

{Assert, {p},
{ store(add, {p}, Assertions, Speaker)
& store(add, {p}, Assertions, Listener) } };

{Concede, {p},
{ if { { inspect(!in, {p}, Concessions, Speaker)
& inspect(in, {p}, Assertions, Listener) }
|| { inspect(!in, {p}, Concessions, Speaker)
& { event(last, Request, {p})
|| event(last, Serious, {p}) } }
then {store(add, {p}, Concessions, Speaker) } } };

{ElementaryArgument, {p, Q},
{ if { inspect(!in, {p}, Concessions, Listener)
& event(past, Challenge, {p}, Listener) }
then { store(add, {p}, Assertions, Speaker)
& store(add, {p}, Concessions, Speaker)
& store(add, {Q}, Assertions, Speaker)
& store(add, {Q}, Concessions, Speaker)
& store(add, <{p}, Q>, Assertions, Speaker)
& store(add, <{p}, Q>, Concessions, Speaker) } } };

{Request, {p},
{ if { inspect(!in, {p}, Concessions, Speaker) }
then { move(mandate, next, Concede, {p})
|| move(mandate, next, WeakRetraction, {p}) } } };

{Serious, {p},
{ if{ inspect(!in, {p}, Dark, Listener)
& { event(last, WeakRetraction, {p})
|| event(last, Challenge, {p}) } }
then { move(mandate, next, Concede, {p})
|| move(mandate, next, WeakRetraction, {p}) } } };

{Resolve, {p, q},
{ if { inspect(in, {p}, Concessions, Listener)
& inspect(in, {q}, Concessions, Listener) }
then { move(mandate, next, WeakRetraction, {p})
|| move(mandate, next, WeakRetraction, {q}) } } };

{Challenge, {p},
{ if { inspect(in, {p}, Assertions, Listener)
& inspect(in, {p}, Concessions, Speaker)
& event(!past, Challenge, {p}, Listener) }
then { move(mandate, next, ElementaryArgument, {p, Q})
|| move(mandate, next, WeakRetraction, {p})
|| move(mandate, next, StrongRetraction, {p}) } } };

{WeakRetraction, {p},
{ if { { inspect(!in, {p}, Dark, Speaker)
& event(!past, Serious, {p}, Listener)
& { event(last, Request, {p}, Listener)
|| event(last, Serious, {p}, Listener) } }
|| { inspect(!in, {p}, Dark, Speaker)
& event(!past, Serious, {p}, Listener)
& inspect(in, {p}, Concessions, Speaker) } }
then { store(remove, {p}, Assertions, Speaker)
& store(remove, {p}, Concessions, Speaker) } } };

{StrongRetraction, {p},
{ store(remove, {p}, Assertions, Speaker) } };

{EndTurn, {p},
{ assign(Speaker, Listener) & assign(Listener, Speaker) } };

}

Similarly, the XML description of the CRG rules was reformulated
into a DGDL description thus:

Unfortunately the CTG DGDL description is quite long and takes
up several pages due to its extensive set of locutions. For this reason,
only a short but representative extract is presented here which in-
cludes the header describing the games components and a selection
of interaction rules:

CRG{
{ turns, magnitude:single, ordering:liberal };
{ players, min:1, max:undefined };
{ player, id:$PlayerID$, role:speaker };

{Initial, scope:initial,
{ move(,next,Suggest1) || move(propose,next,Suggest3)
|| move(propose,next,Suggest6) } };

{Suggest1, {p}, "My idea is", { move(propose,next,Suggest3)
|| move(propose,next,Check6) || move(propose,next,Agree2)
|| move(propose,next,Transform6) || move(propose,next,Agree4) } };

{Suggest2, {p}, "Just imagine", { move(propose,next,Suggest3)
|| move(propose,next,Check6) || move(propose,next,Agree2)
|| move(propose,next,Transform6) || move(propose,next,Agree4) } };

{Suggest3, {p}, "What if", { move(propose,next,Agree3)
|| move(propose,next,Transform6) || move(propose,next,Agree2)
|| move(propose,next,Agree4) || move(propose,next,Check8) } };

{Suggest4, {p}, "How about", { move(propose,next,Check6)
|| move(propose,next,Agree3) || move(propose,next,Transform6)
|| move(propose,next,Agree4) || move(propose,next,Suggest3) } };

{Suggest5, {p}, "I feel", { move(propose,next,Suggest3)
|| move(propose,next,Check6) || move(propose,next,Agree2)
|| move(propose,next,Transform6) || move(propose,next,Suggest3) } };

{Suggest6, {p}, "I think", { move(propose,next,Suggest3)
|| move(propose,next,Check6) || move(propose,next,Agree2)



|| move(propose,next,Transform6) } };
{Suggest7, {p}, "Let me say more about that", { move(propose,next,Question6)
|| move(propose,next,Check6)
|| move(propose,next,Agree5) || move(propose,next,Agree1) } };

{Suggest8, {p}, "An example", { move(propose,next,Check6)
|| move(propose,next,Check8) || move(propose,next,Transform7)
|| move(propose,next,Agree5) || move(propose,next,Agree1) } };

{Question1, {p}, "Why?", { move(propose,next,Suggest7)
|| move(propose,next,Transform6) || move(propose,next,Suggest5) } };

{Question2, {p}, "Can you say more on that?", { move(propose,next,Suggest7)
|| move(propose,next,Suggest5) } };

{Question3, {p}, "Does this connect with anything for you?",
{ move(propose,next,Suggest5) || move(propose,next,Transform2)
|| move(propose,next,Maintain2) } };

{Question4, {p}, "What do you mean when you say?", { move(propose,next,Suggest7)
|| move(propose,next,Transform6) } };

{Question5, {p}, "Why do you think that?", { move(propose,next,Suggest6) } };
{Question6, {p}, "Why do you feel that?", { move(propose,next,Suggest5)
|| move(propose,next,Suggest6)
|| move(propose,next,Suggest7) } };

{Question7, {p}, "What are the possible alternatives?",
{ move(propose,next,Transform6) || move(propose,next,Check4) } };

{Question8, {p}, "Has anyone got another idea?",
{ move(propose,next,Suggest1) || move(propose,next,Suggest4)
|| move(propose,next,Suggest3) || move(propose,next,Suggest6) } };

}

The complete reformulation of CRG is available however along
with a description of PPD0 from the DGDL repository5.

4 MAPPING DIALOGUES ONTO
ARGUMENTATION STRUCTURES

To adopt InterLoc as an interface to the WWAW, arguments ex-
pressed in CRG dialogues must be mapped onto AIF and ideally the
process by which this occurs should be automated so that the process
of taking a CRG dialogue, extracting the arguments, and transcrib-
ing them into AIF, is low cost and doesn’t require transcription by
experts. Similarly for PPD0, argument structures must be extracted
from the PPD0 dialogues and recorded in AIF.

The initial approach has been to map particular dialogical se-
quences onto argumentation structures which are subsequently ex-
pressed as, and recorded as AIF. For example, in the sequence
Assertion–Challenge–Elementary Argument from PPD0, the initial
assertion can be interpreted as a position taken by one player upon an
issue. The Elementary Argument then provides premises in support
of the Assertion. However this does not happen in isolation within
a dialogue, that the player provides a conclusion and supporting ar-
gument and that they are related utterances, but arises as a result of
the dialogical interaction between the players due to the intervening
Challenge move. This kind of sequence can be consistently mapped
onto an AIF argument structure providing a conclusion and support-
ing premises, licensed by an intermediate, possibly undefined, argu-
mentation scheme. Similarly in CRG, the sequence Suggest-Check-
Suggest can be consistently mapped onto an AIF argument.

The following four examples explore simple dialogues on the
regulation of financial institutions in both PPD0 and CRG and
map the underlying arguments onto AIF which are then compared.
Dialogues I and II illustrate the Assertion–Challenge–Elementary
Argument and Suggest-Check-Suggest mappings discussed earlier.

Dialogue I: PPD0

1. B: I think there should be greater regulation of financial institu-
tions (assert)
2. W: I’m not so sure (challenge)
3. B: Let me say more; by increasing regulation, we reduce the
chance of repeating recent problems (elementary argument)

Dialogue II: CRG
1. B: I think there should be greater regulation of financial institu-
tions (suggest6)
2. W: I’m not so sure (check6)

5 http://www.arg.computing.dundee.ac.uk/projects/a4a/dgdl/repository/

3. B: Let me say more; by increasing regulation, we reduce the
chance of repeating recent problems (suggest 7)

It is of interest that in dialogues I and II the AIF representation
of the arguments, illustrated in figure 1, is identical even though the
dialogues were generated according to the rules of different games.
In this case the AIF enables us to easily compare the arguments ex-
pressed in the dialogues.

Dialogues III and IV again illustrate interactions from PPD0 and
CRG that exhibit surface similarity in the actual utterances of the
players but yield different AIF mappings as a result of their differing
underlying dialogue game protocols.

Dialogue III: CRG
1. B: I think there should be greater regulation of financial institu-
tions (suggest6)
2. W: I’m not so sure (check6)
3. B: So what I think you are saying is that we should not regulate at
all, that’s crazy! (transform1)
4. W: No, that’s not what I’m saying (maintain2)

Dialogue IV: PPD0

1. B: I think there should be greater regulation of financial institu-
tions (assert)
2. W: I’m not so sure (challenge)
3. B: Are you serious that you don’t think there should be more
regulation? (extractor serious?)
4. W: I am not committed to there being more regulation! (weak
retraction)

In dialogues III and IV, although the locutors have expressed simi-
lar things, the commitment model of PPD0 results in a very different
AIF representation of the arguments expressed at that stage in the
dialogue as shown in figure 1. In the AIF representation of dialogue
III, the argument expressed in the dialogue fragment yields a similar
structure to that from dialogues I and II but the conclusion is war-
ranted by a different argumentation scheme. The players end up in
conflict causing the introduction of a conflict scheme. Although the
source dialogues are superficially similar, the AIF generated from
dialogue IV is very different, the retraction at turn 4 causes only the
I-node for the content of turn 1 to remain. It should be noted that
it appears as though in dialogue IV, arguments introduced into the
dialogue by player B are being disregarded because they no longer
appear in the AIF representation of the argument, at that point during
the dialogue. This is a result of the underlying commitment model of
PPD0 which demonstrates how different arguments can be produced
as a result of commitment rules, altough the locutions associated with
those rules are superficially similar.

The example dialogues demonstrate clear differences between the
two dialogue games, CRG and PPD0. The rules of each game are dif-
ferent and this can be verified through visual inspection both of the
original rules and also through comparison of the reformulations into
DGDL. These games also yield different dialogues, because certain
chains of locutions that are legal in a more expressive game like CRG
are either prohibited or not possible in a more structured game like
PPD0. However, a further complication occurs in that the particular
rules of an individual game can result in similar dialogues but dif-
ferent underlying argumentation structures once those dialogues are
analysed for their argumentative content as demonstrated by the AIF
diagrams of the example dialogues.

Figure 1. Fragments of AIF diagrams from the example dialogues.
Dialogues I and II, produced from PPD0 and CRG respectively, yield the
same underlying AIF argumentation structure from superficially similar

dialogues. Dialogues III and IV, whilst superficially similar at the dialogue
level, yield different underlying AIF argumentation structures due to the

effect of the commitment model of PPD0 in dialogue IV

5 CONCLUSIONS & FURTHER WORK
It is clear that the expressiveness and naturalness of CRG dialogues
contributes to user acceptance of InterLoc, an attribute that would be
advantageous if InterLoc is to be used as an interface to the WWAW.
However, more restrictive dialogue games like PPD0 can prove to
be better protocols for autonomous agent communication if WWAW
resources are to be used to provide knowledge-bases for agents inter-
acting within InterLoc dialogues. An ideal solution would be there-
fore to select and integrate elements of both games, balancing the
expressiveness and naturalness of CRG against the argumentative
rigour of PPD0.

Our future work will therefore explore variant CRG games that in-
corporate commitment models to make the dialogues more tractable
for agents whilst retaining the flexibilty of the current CRG ruleset.
This will enable us to pursue the twin goals of adopting InterLoc as
a WWAW interface whilst enabling InterLoc users to interact with
existing WWAW resources.

